Fourier Serie Expansion (Other) Example

Triangular wave

```
> restart:
> S := unapply( (1/2)*sum( c(k)*exp(-I*k*t), k=-N..N), N) ;
```

$$S := N \to \frac{1}{2} \sum_{k=-N}^{N} piecewise\left(k = 0, \pi, ck\right)(k)\, e^{-I\,k\,t} \tag{1.1}$$

```
> f := piecewise( t<0, Pi+t, Pi-t ) ;
```

$$f := \begin{cases} \pi + t & t < 0 \\ \pi - t & otherwise \end{cases} \tag{1.2}$$

```
> plot( f, t=-Pi..Pi ) ;
```




```
> c0 := int( f, t=-Pi..Pi ) / Pi ;
```

$$c0 := \pi \tag{1.3}$$

```
> int( f*exp(-I*k*t), t=-Pi..Pi ) / Pi ; simplify(%) ;
```

```
ck := unapply( simplify( % ), k ) assuming k::integer ;
```

$$\frac{\dfrac{-e^{I k \pi} + I k \pi + 1}{k^2} - \dfrac{\left(I e^{I k \pi} k \pi - e^{I k \pi} + 1\right) e^{-I k \pi}}{k^2}}{\pi}$$

$$-\frac{2\left(\cos(k\pi) - 1\right)}{k^2 \pi}$$

$$ck := k \to \frac{2\left((-1)^{k+1} + 1\right)}{k^2 \pi} \tag{1.4}$$

```
> c := piecewise( k=0, c0, ck ) ;
```

$$c := \begin{cases} \pi & k = 0 \\ ck & otherwise \end{cases} \tag{1.5}$$

```
> S(5) ; simplify(%) ;
```

$$\frac{2}{25}\frac{e^{5 I t}}{\pi} + \frac{2}{9}\frac{e^{3 I t}}{\pi} + \frac{2 e^{I t}}{\pi} + \frac{1}{2}\pi(0) + \frac{2 e^{-I t}}{\pi} + \frac{2}{9}\frac{e^{-3 I t}}{\pi} + \frac{2}{25}\frac{e^{-5 I t}}{\pi}$$

$$\frac{1}{450}\frac{-640\cos(t)^3 + 1560\cos(t) + 1152\cos(t)^5 + 225\pi^2}{\pi} \tag{1.6}$$

```
> plot( [f,simplify(S(101))], t=-Pi..Pi ) ;
```



Square wave

```
> restart:
> S := unapply( (1/2)*sum( c(k)*exp(I*k*t), k=-N..N), N) ;
```

$$S := N \to \frac{1}{2} \sum_{k=-N}^{N} c(k)\, e^{I k t} \tag{2.1}$$

```
> f := piecewise( t<0, -1, 1 ) ;
```

$$f := \begin{cases} -1 & t < 0 \\ 1 & otherwise \end{cases} \tag{2.2}$$

```
> plot( f, t=-Pi..Pi ) ;
```




```
> c0 := int( f, t=-Pi..Pi ) / Pi ;
```

$$c0 := 0 \tag{2.3}$$

```
> int( f*exp(-I*k*t), t=-Pi..Pi ) / Pi ; simplify(expand(%)) ;
  ck := unapply( simplify( % ), k ) assuming k::integer ;
```

$$\frac{\frac{I\left(e^{I\pi k}-1\right)}{k}-\frac{I\left(e^{I\pi k}-1\right)e^{-I\pi k}}{k}}{\pi}$$

$$\frac{2I\left(\cos(\pi k)-1\right)}{\pi k}$$

$$ck := k \rightarrow \frac{2I\left((-1)^k-1\right)}{\pi k} \tag{2.4}$$

```
> c := piecewise( k=0, c0, ck ) ;
```

$$c := \begin{cases} 0 & k = 0 \\ ck & otherwise \end{cases} \tag{2.5}$$

```
> S(5) ; simplify(%) ;
```

$$\frac{\frac{2}{5}\,\mathrm{I}\,e^{-5\,\mathrm{I}\,t}}{\pi}+\frac{\frac{2}{3}\,\mathrm{I}\,e^{-3\,\mathrm{I}\,t}}{\pi}+\frac{2\,\mathrm{I}\,e^{-\mathrm{I}\,t}}{\pi}-\frac{2\,\mathrm{I}\,e^{\mathrm{I}\,t}}{\pi}-\frac{\frac{2}{3}\,\mathrm{I}\,e^{3\,\mathrm{I}\,t}}{\pi}-\frac{\frac{2}{5}\,\mathrm{I}\,e^{5\,\mathrm{I}\,t}}{\pi}$$

$$\frac{4}{15}\,\frac{5\sin(3\,t)+3\sin(5\,t)+15\sin(t)}{\pi} \qquad (2.6)$$

```
> plot( [f,simplify(S(101))], t=-Pi..Pi, y=-1.2..1.2 ) ;
```

Half sin

```
> restart:
> S := unapply( (1/2)*sum( c(k)*exp(I*k*t), k=-N..N), N) ;
```

$$S := N \to \frac{1}{2}\sum_{k=-N}^{N} c(k)\,e^{\mathrm{I}\,k\,t} \qquad (3.1)$$

```
> f := piecewise( t<0, 0, sin(t) ) ;
```

$$f := \begin{cases} 0 & t < 0 \\ \sin(t) & otherwise \end{cases} \qquad (3.2)$$

```
> plot( f, t=-Pi..Pi ) ;
```



```
> c0 := int( f, t=-Pi..Pi ) / Pi ;
```

$$c0 := \frac{2}{\pi} \tag{3.3}$$

```
> int( f*exp(-I*t), t=-Pi..Pi ) / Pi ;
  c1 := unapply( simplify( % ), k ) assuming k::integer ;
```

$$-\frac{1}{2}\,\mathrm{I}$$

$$c1 := k \to -\frac{1}{2}\,\mathrm{I} \tag{3.4}$$

```
> int( f*exp(I*t), t=-Pi..Pi ) / Pi ;
  cm1 := unapply( simplify( % ), k ) assuming k::integer ;
```

$$\frac{1}{2}\,\mathrm{I}$$

$$cm1 := k \to \frac{1}{2}\,\mathrm{I} \tag{3.5}$$

```
> int( f*exp(-I*k*t), t=-Pi..Pi ) / Pi ;
  ck := unapply( simplify( % ), k ) assuming k::integer ;
```

$$-\frac{\left(1+\mathrm{e}^{\mathrm{I}\pi k}\right)\mathrm{e}^{-\mathrm{I}\pi k}}{\left(k^2-1\right)\pi}$$

$$ck := k \to -\frac{1+(-1)^k}{\left(k^2-1\right)\pi} \tag{3.6}$$

```
> c := piecewise( k=0, c0, k=1, c1, k=-1, cm1, ck ) ;
```

$$c := \begin{cases} \frac{2}{\pi} & k = 0 \\ c1 & k = 1 \\ cm1 & k = -1 \\ ck & otherwise \end{cases} \tag{3.7}$$

```
> plot( [f,simplify(S(10))], t=-Pi..Pi, y=-1.2..1.2 ) ;
```



Triangular wave (with cos and phase)

```
> restart:
> S := unapply( 'a0/2 + sum( A(k)*cos(k*t-phi(k)), k=1..N)', N) ;
```

$$S := N \rightarrow \frac{1}{2}\, a0 + \sum_{k=1}^{N} A(k)\, \cos\left(k\, t - \phi(k)\right) \tag{4.1}$$

```
> f := piecewise( t<0, Pi+t, Pi-t^2/Pi ) ;
```

$$f := \begin{cases} \pi + t & t < 0 \\ \pi - \dfrac{t^2}{\pi} & otherwise \end{cases} \tag{4.2}$$

```
> plot( f, t=-Pi..Pi ) ;
```




```
> a0 := int( f, t=-Pi..Pi ) / Pi ;
```

$$a0 := \frac{7}{6}\,\pi \tag{4.3}$$

```
> int( f*cos(k*t), t=-Pi..Pi ) / Pi ;
  ak := unapply( simplify( % ), k ) assuming k::integer ;
```

$$\frac{-\frac{\cos(k\pi)-1}{k^2} - \frac{2\left(\cos(k\pi)\,k\pi-\sin(k\pi)\right)}{\pi\,k^3}}{\pi}$$

$$ak := k \to \frac{3\,(-1)^{k+1}+1}{k^2\,\pi} \tag{4.4}$$

```
> int( f*sin(k*t), t=-Pi..Pi ) / Pi ;
  bk := unapply( simplify( % ), k ) assuming k::integer ;
```

$$\frac{-\frac{-\sin(k\pi)+k\pi}{k^2} + \frac{\pi^2\,k^2+2-2\sin(k\pi)\,k\pi-2\cos(k\pi)}{\pi\,k^3}}{\pi}$$

(4.5)

$$bk := k \to \frac{2\left((-1)^{k+1}+1\right)}{\pi^2 k^3} \tag{4.5}$$

```
> A := unapply( simplify(sqrt(ak(k)^2+bk(k)^2)), k) assuming
  k::integer ;
```

$$A := k \to \frac{\sqrt{2}\sqrt{5\pi^2 k^2 + 3\pi^2 k^2 (-1)^{k+1} + 4 + 4(-1)^{k+1}}}{\pi^2 |k|^3} \tag{4.6}$$

```
> phi := unapply( simplify(arctan( bk(k), ak(k))), k ) assuming
  k::integer, k>0;
```

$$\phi := k \to \arctan\left(-\frac{2\left((-1)^k - 1\right)}{\pi^2 k^3}, \frac{3(-1)^{k+1}+1}{k^2 \pi}\right) \tag{4.7}$$

```
> S(5) ;
```

$$\frac{7}{12}\pi + \frac{\sqrt{2}\sqrt{8\pi^2+8}\cos\left(t - \arctan\left(\frac{1}{\pi}\right)\right)}{\pi^2} - \frac{1}{8}\frac{\sqrt{2}\sqrt{8}\sqrt{\pi^2}\cos(2t)}{\pi^2} \tag{4.8}$$

$$+ \frac{1}{27}\frac{\sqrt{2}\sqrt{72\pi^2+8}\cos\left(3t - \arctan\left(\frac{1}{3\pi}\right)\right)}{\pi^2}$$

$$- \frac{1}{64}\frac{\sqrt{2}\sqrt{32}\sqrt{\pi^2}\cos(4t)}{\pi^2}$$

$$+ \frac{1}{125}\frac{\sqrt{2}\sqrt{200\pi^2+8}\cos\left(5t - \arctan\left(\frac{1}{5\pi}\right)\right)}{\pi^2}$$

```
> plot( [f,S(101)], t=-Pi..Pi ) ;
```


3
2
1
$-\pi$
$-\frac{3\pi}{4}$
$-\frac{\pi}{2}$
$-\frac{\pi}{4}$
0
$\frac{\pi}{4}$
$\frac{\pi}{2}$
$\frac{3\pi}{4}$
π
t